

Charles Patridge PDPC, Ltd.

172 Monce Rd.

Burlington, CT 06013 USA

Phone: 860-673-9278



WWW.SCONSIG.COM

Email: Charles_S_Patridge@prodigy.net

The Power of Proc Format

- Minimize SAS Code
- Easier Maintenance of SAS Code
- Easier SAS Code to Understand
- Potentially Faster than SQL

Exit for Video

How to Create Permanent Formats

```
libname sasuser "C:\PROC_FORMAT";  
libname myfmts "C:\PROC_FORMAT";  
  
data myfmts.states_region;  
length states $2. region $2. description $20.;  
infile "C:\PROC_FORMAT\states_region.csv"  
       delimiter=' ' firstobs=2 missover;;  
input states region description ;  
run;
```

How to Create Permanent Formats

```
State,Region,Description
AL,SE,SouthEast
AR,CS,CentralSouth
AZ,SW,SouthWest
CA,SW,SouthWest
CO,SW,SouthWest
CT,NE,NorthEast
DC,MA,MidAtlantic
DE,MA,MidAtlantic
FL,SE,SouthEast
GA,SE,SouthEast
IA,MN,MidWest
ID,NW,NorthWest
IL,CN,CentralNorth
IN,CN,CentralNorth
KS,CS,CentralSouth
KY,CN,CentralNorth
LA,CS,CentralSouth
:
:
```

Here is what the
States_Region.csv file looks like

How to Create Permanent Formats

```
data myfmts.states_region_fmt (keep=fmtname start label);
retain fmtname '$regfmt';
length start $5. ;
set myfmts.states_region end=eof;
start = states;
label = region;
output;
if eof then do;
    start = 'OTHER';
    label = '';
    output;
end;
run;
```

How to Create Permanent Formats

```
proc format lib=myfmts cntlin=myfmts.states_region_fmt ;  
run;
```

```
proc format lib=myfmts;  
  value $statfmt  
    ' ' = "TOTAL";  
run;
```

```
options ls=80 center nodate;  
proc format lib=myfmts page; run;
```


How to Create Permanent Formats

```
libname sasuser "C:\PROC_FORMAT";
```

```
libname myfmts "C:\PROC_FORMAT";
```

```
options fmtsearch=(work library myfmts);
```

How to Create Permanent Formats

```
/** Get, Select, and Filter Data */
/** create temp SAS dataset called Select */
Data Select;
/** read permanent SAS dataset SASALARY from Library SASUSER */
set sasuser.SASALARY;

/** Select Full Time Employees */
if "FULL TIME EMPLOYEE" = upcase( empltype );

/** get state code */
state = zipstate ( substr( left( zipcode ), 1, 5 ) );

/** create Region code using a Format */
region = put( state, $regfmt.);

/** selection filter - get NE only */
if region = 'NE';
Title1 "New England - Statistics";
run;
```

Compare Format to SQL

```
proc sql noprint;
create table SASALARY_TEMP as
  select *
    from sasuser.sasalary as m
      left join
        myfmts.states_region as s
    on s.states = zipstate ( substr( left( m.zipcode ), 1, 5 ) );
quit;
```

Left Join 2 Tables
to acquire Region
Code

Data Select;

```
/** read temporary SAS dataset SASALARY_TEMP */
set SASALARY_TEMP;

/** Select Full Time Employees */
if "FULL TIME EMPLOYEE" = upcase( empctype );

/** selection filter - get NE only */
if region = 'NE';
Title1 "New England - Statistics";
run;
```

Compare Format to SQL- CPU Stats

Format

```
10      Data Select;
11      set sasuser.SASALARY;
12
13
14      if "FULL TIME EMPLOYEE" = upcase( empltype );
15
16      state=zipstate(substr( left(zipcode ),1,5) );
17
18      region = put( state, $regfmt.);
19
20
21      if region = 'NE';
22      Title1 "New England - Statistics";
23
24      run;
```

NOTE: Invalid argument to function ZIPSTATE at line 18 column 11.

NOTE: There were 1392 observations read from the data set SASUSER.SASALARY.

NOTE: The data set WORK.SELECT has 99 observations and 21 variables.

NOTE: DATA statement used (Total process time):

real time	0.15 seconds	
user cpu time	0.00 seconds	
system cpu time	0.07 seconds	
Memory		185k

SQL

```
9      proc sql noprint;
10      create table SASALARY_TEMP as
11      select *
12      from sasuser.sasalary      as m
13      left join
14      myfmts.states_region as s
15      on
16      s.states=zipstate(substr(left(m.zipcode),1,5));
17
18      NOTE: Invalid argument to function ZIPSTATE. Missing values
19      may be NOTE: Table WORK.SASALARY_TEMP created, with 1392
20      rows and 22 columns.
21
22      NOTE: PROCEDURE SQL used (Total process time):
23
24      real time          0.26 seconds
25      user cpu time      0.03 seconds
26      system cpu time   0.15 seconds
27      Memory              2945k
```

```
21      Data Select;
22      set          SASALARY_TEMP;
23
24      if "FULL TIME EMPLOYEE" = upcase( empltype );
25
26      if region = 'NE';
27
28      Title1 "New England - Statistics";
29
30      run;
```

NOTE: There were 1392 observations read from the data set WORK.SASALARY_TEMP.

NOTE: The data set WORK.SELECT has 99 observations and 22 variables.

NOTE: DATA statement used (Total process time):

real time	0.03 seconds	
user cpu time	0.00 seconds	
system cpu time	0.03 seconds	
Memory		185k

The
End

Charles Patridge

PDPC, Ltd.

172 Monce Rd.

Burlington, CT 06013 USA

Phone: 860-673-9278



WWW.SCONSIG.COM

Email: Charles_S_Patridge@prodigy.net